Puzzle Corner, July/August 2010-2, 8x8 Sudoku-like problem

Joel Karnofsky, 7/16/10

To fill in an 8x8 grid with digits 1-8 so no digit repeats horizontally, vertically or in a non-wrapping diagonal, with as few holes as possible.

In the following I treat the 8x8 grid as a linear array. A cell value of '0' means not yet assigned. 'rows' is a list of lists of indices defining a row.

```
In[1]:= dim = 8;
        data1 = Array[0 &, {dim dim}];
        rows = Partition[Range[dim dim], dim];
```

Precompute all the horizontal, vertical and diagonal cells associated with a given one. 'linked1[[i]]' is a list of indices associated to cell 'i'.

```
In[4]:= toOne[i_, j_] := i dim - dim + j;
        linked[i_, j_] := DeleteCases[Union[
            Table[{k, j}, {k, dim}],
            Table[{i, k}, {k, dim}],
            Table[{i + j - k, k}, {k, Max[1, i + j - dim], Min[dim, i + j - 1]}],
            Table[{i - j + k, k}, {k, Max[1, j - i + 1], Min[dim, j - i + dim]}]
          ], {i, j}];
        linked1 = Flatten[Table[toOne @@@ linked[i, j], {i, dim}, {j, dim}], 1];
```

Recursive function that generates all acceptable patterns of '1's, using that there must be exactly one '1' in each row.

```
In[7]:= next2[{}] := res = Append[res, data1];
        next2[{row_, more___}] := Scan[
          If[FreeQ [data1[[linked1[[#]]]], 1, {1}],
            data1[[#]] = 1; next2[{more}]; data1[[#]] = 0
           ] &,
          row
         ];
```

Functions to compute a unique representation of a pattern allowing for all 8 symmetries of a square and all permutations of the non-zero digits assigned so far

```
In[9]:= rot[data_] := Transpose[Reverse /@ data];
        sym[data1_] := Module[{data = Partition[data1, dim]},
          Flatten[#, 1] & /@ Join[
            NestList[rot@# &, data, 3],
            NestList[rot@# &, Transpose@data, 3]]
         ];
        minimal[list_] := list[[First@Ordering[list, 1]]];
        unique[data11_] := minimal@Flatten[sym[Prepend[#, 0][[data11]]] & /@ perms, 1];
```

Functions to add a pattern for a new digit in all possible non-overlapping ways to all existing patterns, keeping only one representative from each symmetry//permutation class.

```
comb[old_, new_] := Flatten[Last@Reap[
     Scan[Function[one, Scan[If[one.# == 0, Sow[unique[one + # + 1]]] &, new]], old]
    ], 1];
two[old_, new_] := Block[{perms = Permutations@Range[++digit]},
   Union@comb[old, new /. 1 → digit]
  ];
```

This run shows that there are no valid ways to fill in 7 digits with 8 copies of each. Hence it is not possible to find solutions to the original problem with 8 digits leaving either one or no holes.

```
In[15]:= Timing[
  res = {}; next2[rows];
  digit = 1; Nest[two[#, res] &, res, 6]
 ]

Out[15]= {17.535, {}}
```

This run finds 12 unique solutions with two holes. Using the above, this can only occur when 6 digits have 8 copies and 2 have 7.

```
In[16]:= Timing[
  res = {}; next2[rows];
  digit = 1; res6 = Nest[two[#, res] &, res, 5];
  res = {}; Do[next2[Delete[rows, i]], {i, dim}];
  (res8 = Nest[(two[#, res]) &, res6, 2]) // Length
 ]

Out[16]= {124.552, 12}
```

The 12 results with two holes (zeros).

In[17]:=
```
MatrixForm@Partition[#, dim] & /@ res8
```

Out[17]=

$$\left\{\begin{pmatrix} 0&1&2&3&4&5&6&0\\ 7&4&6&5&2&1&3&8\\ 1&2&7&4&3&8&5&6\\ 5&6&3&8&7&4&1&2\\ 3&8&5&1&6&2&7&4\\ 6&7&4&2&5&3&8&1\\ 4&5&1&7&8&6&2&3\\ 2&3&8&6&1&7&4&5 \end{pmatrix}, \begin{pmatrix} 0&1&2&3&4&5&6&0\\ 7&4&6&5&2&1&3&8\\ 1&2&8&4&3&7&5&6\\ 5&6&3&7&8&4&1&2\\ 3&7&5&1&6&2&8&4\\ 6&8&4&2&5&3&7&1\\ 4&5&1&8&7&6&2&3\\ 2&3&7&6&1&8&4&5 \end{pmatrix}, \begin{pmatrix} 0&1&2&3&4&5&6&7\\ 2&6&7&5&1&8&3&4\\ 7&5&8&4&3&6&2&1\\ 4&3&6&1&2&7&8&5\\ 6&7&4&8&5&3&1&2\\ 3&2&0&6&7&4&5&8\\ 5&4&3&2&8&1&7&6\\ 1&8&5&7&6&2&4&3 \end{pmatrix}, \begin{pmatrix} 0&1&2&3&4&5&6&7\\ 2&6&7&5&8&1&3&4\\ 7&5&1&4&3&6&2&8\\ 4&3&6&8&2&7&1&5\\ 6&7&4&0&5&3&8&2\\ 3&2&8&6&7&4&5&1\\ 5&4&3&2&1&8&7&6\\ 1&8&5&7&6&2&4&3 \end{pmatrix},\right.$$

$$\begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&2&7&1&4&6\\ 8&6&0&5&4&0&3&1\\ 7&4&1&6&3&8&5&2\\ 6&8&5&7&2&4&1&3\\ 4&3&2&1&8&7&6&5\\ 2&1&4&3&6&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&2&7&1&4&6\\ 8&6&0&5&4&2&3&1\\ 7&4&1&6&3&8&5&2\\ 6&8&5&7&2&4&1&3\\ 4&3&0&1&8&7&6&5\\ 2&1&4&3&6&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&2&7&1&4&6\\ 8&6&7&5&4&2&3&1\\ 7&4&1&6&3&8&5&2\\ 6&8&5&7&2&4&1&3\\ 4&3&0&1&8&0&6&5\\ 2&1&4&3&6&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&6&7&1&4&2\\ 8&6&0&5&4&0&3&1\\ 7&4&1&2&3&8&5&6\\ 2&8&5&7&6&4&1&3\\ 4&3&2&1&8&7&6&5\\ 6&1&4&3&2&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix},$$

$$\left.\begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&6&7&1&4&2\\ 8&6&0&5&4&2&3&1\\ 7&4&1&2&3&8&5&6\\ 2&8&5&7&6&4&1&3\\ 4&3&0&1&8&7&6&5\\ 6&1&4&3&2&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&5&8&6&7&1&4&2\\ 8&6&7&5&4&0&3&1\\ 7&4&1&2&3&8&5&6\\ 2&8&5&7&6&4&1&3\\ 4&3&2&1&8&0&6&5\\ 6&1&4&3&2&5&8&7\\ 5&7&6&8&1&3&2&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&7&8&6&1&2&4&5\\ 8&6&2&5&4&7&3&1\\ 5&4&7&0&3&8&2&6\\ 7&8&5&0&6&4&1&3\\ 4&3&1&7&8&5&6&2\\ 6&5&4&3&2&1&8&7\\ 2&1&6&8&7&3&5&4 \end{pmatrix}, \begin{pmatrix} 1&2&3&4&5&6&7&8\\ 3&7&8&6&2&1&4&5\\ 8&6&0&5&4&7&3&2\\ 5&4&7&2&3&8&1&6\\ 7&8&5&1&6&4&2&3\\ 4&3&0&7&8&5&6&1\\ 6&5&4&3&1&2&8&7\\ 2&1&6&8&7&3&5&4 \end{pmatrix}\right\}$$

The 12 results in color, holes in black.

In[18]:=
```
ArrayPlot[Partition[#, dim], Mesh → True, ColorRules → {0 → Black},
   ColorFunction → "LightTemperatureMap", ImageSize → 135] & /@ res8
```

Out[18]=